
 
Projekt:   Erklärung der Simulationsfunktion 
 
Programm:   %17000 
 
Programmierung:  Heidenhain iTNC (Klartext) 

 
 
0  BEGIN PGM 17000 MM  
1  BLK FORM 0.1 Z  X-20  Y-32  Z-60 
2  BLK FORM 0.2 IX+40 IY+64 IZ+60 
3  TOOL CALL 22 Z S6000 F600 ;SFR VHM D = 20 mm 
4  M3 
5  L  X+70  Y+0 R0 FMAX 
6  L  Z+2 R0 FMAX 
7  L  Z+0 R0 F AUTO 
8  LBL 1 
9  L IZ-15 R0 F AUTO 
10 APPR LCT  X+20  Y+0 R5 RL 
11 L  Y-24 
12 L  X-20 
13 L  Y+20 
14 L  X+20 
15 L  Y+0 
16 DEP LCT  X+70  Y+0 R5 
17 LBL 0 
18 CALL LBL 1 REP2 
19 L  Z+50 R0 FMAX 
20 TOOL CALL 9 Z S6000 F500 ;SFR VHM D = 8 mm 
21 M3 
22 L  X+0  Y+0 R0 FMAX 
23 L  Z+2 R0 FMAX 
24 CYCL DEF 252 KREISTASCHE ~ 
    Q215=+0    ;BEARBEITUNGS-UMFANG ~ 
    Q223=+32.1 ;KREISDURCHMESSER ~ 
    Q368=+0.5  ;AUFMASS SEITE ~ 
    Q207= AUTO ;VORSCHUB FRAESEN ~ 
    Q351=+1    ;FRAESART ~ 
    Q201=-3.6  ;TIEFE ~ 
    Q202=+3.6  ;ZUSTELL-TIEFE ~ 
    Q369=+0.5  ;AUFMASS TIEFE ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q338=+1    ;ZUST. SCHLICHTEN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q203=+0    ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. ~ 
    Q370=+1    ;BAHN-UEBERLAPPUNG ~ 
    Q366=+1    ;EINTAUCHEN ~ 
    Q385= AUTO ;VORSCHUB SCHLICHTEN 
25 CYCL CALL 
26 CYCL DEF 252 KREISTASCHE ~ 
    Q215=+0    ;BEARBEITUNGS-UMFANG ~ 
    Q223=+28   ;KREISDURCHMESSER ~ 
    Q368=+0.4  ;AUFMASS SEITE ~ 
    Q207= AUTO ;VORSCHUB FRAESEN ~ 
    Q351=+1    ;FRAESART ~ 
    Q201=-44   ;TIEFE ~ 
    Q202=+10   ;ZUSTELL-TIEFE ~ 



 
    Q369=+0.5  ;AUFMASS TIEFE ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q338=+1    ;ZUST. SCHLICHTEN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q203=+0    ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. ~ 
    Q370=+1    ;BAHN-UEBERLAPPUNG ~ 
    Q366=+1    ;EINTAUCHEN ~ 
    Q385= AUTO ;VORSCHUB SCHLICHTEN 
27 CYCL CALL 
28 CC  X+0  Y+0 
29 L  X+45  Y+0 R0 FMAX 
30 L  Z-6 R0 F AUTO 
31 APPR LCT  X+17.5  Y+0 R5 RR 
32 C  X+17.5  Y+0 DR+ 
33 DEP LCT  X+45  Y+0 R5 
34 L  Z+0 R0 F AUTO 
35 LBL 2 
36 L IZ-15 R0 F AUTO 
37 APPR LCT  X+19  Y+0 R5 RR 
38 C  X-13  Y-13.856 DR+ 
39 RND R5 
40 L  X-13  Y-22 
41 RND R1 
42 L  X+13 
43 RND R1 
44 L  X+13  Y-13.856 
45 RND R5 
46 C  X+19  Y+0 DR+ 
47 DEP LCT  X+45  Y+0 R5 
48 LBL 0 
49 CALL LBL 2 REP2 
50 L IZ-7.6 
51 APPR LCT  X+19  Y+0 R5 RR 
52 CC  X-151  Y+0 
53 C  X+16.15  Y+31 DR+ 
54 RND R5 
55 L  X-16.15 
56 RND R5 
57 CC  X+151  Y+0 
58 C  X-16.15  Y-31 DR+ 
59 RND R5 
60 L  X+16.15 
61 RND R5 
62 CC  X-151  Y+0 
63 C  X+19  Y+0 DR+ 
64 DEP LCT  X+45  Y+0 R5 
65 L  Z+50 R0 FMAX 
66 TOOL CALL 27 Z S4000 F400 ;SPIBO D = 6.8 mm 
67 M3 
68 CYCL DEF 205 UNIVERSAL-TIEFBOHREN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q201=-10   ;TIEFE ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q202=+5    ;ZUSTELL-TIEFE ~ 
    Q203=-45   ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. ~ 
    Q212=+0    ;ABNAHMEBETRAG ~ 
    Q205=+0    ;MIN. ZUSTELL-TIEFE ~ 
    Q258=+0.2  ;VORHALTEABSTAND OBEN ~ 
    Q259=+0.2  ;VORHALTEABST. UNTEN ~ 



 
    Q257=+0    ;BOHRTIEFE SPANBRUCH ~ 
    Q256=+0.2  ;RZ BEI SPANBRUCH ~ 
    Q211=+0    ;VERWEILZEIT UNTEN ~ 
    Q379=+0    ;STARTPUNKT ~ 
    Q253=+750  ;VORSCHUB VORPOS. 
69 LBL 3 
70 CYCL CALL POS  X+11.939  Y-26  Z+0 
71 CYCL CALL POS  X-11.939  Y-26  Z+0 
72 CYCL CALL POS  X-11.939  Y+26  Z+0 
73 CYCL CALL POS  X+11.939  Y+26  Z+0 
74 LBL 0 
75 TOOL CALL 28 Z S5000 F500 ;SFR VHM D = 7 mm 
76 M3 
77 CYCL DEF 200 BOHREN ~ 
    Q200=+2    ;SICHERHEITS-ABST. ~ 
    Q201=-2.375;TIEFE ~ 
    Q206=+150  ;VORSCHUB TIEFENZ. ~ 
    Q202=+5    ;ZUSTELL-TIEFE ~ 
    Q210=+0    ;VERWEILZEIT OBEN ~ 
    Q203=-45   ;KOOR. OBERFLAECHE ~ 
    Q204=+50   ;2. SICHERHEITS-ABST. ~ 
    Q211=+0.5  ;VERWEILZEIT UNTEN 
78 CALL LBL 3 
79 END PGM 17000 MM  
 
 


